
instructor_ex
    

Structured, Ecto outputs with OpenAI (and OSS LLMs)

Instructor.ex is a spiritual port of the great Instructor Python Library by @jxnlco, check out his talk on youtube.
This library brings structured prompting to LLMs. Instead of receiving text as output, Instructor will coax the LLM to output valid JSON that maps directly to the provided Ecto schema.
If the LLM fails to do so, or provides values that do not pass your validations, it will provide you utilities to automatically retry with the LLM to correct errors.
By default it's designed to be used with the OpenAI API, however it provides an extendable adapter behavior to work with ggerganov/llama.cpp and Bumblebee (Coming Soon!).
At its simplest, usage is pretty straightforward,
defmodule SpamPrediction do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  ## Field Descriptions:
  - class: Whether or not the email is spam
  - reason: A short, less than 10 word rationalization for the classification
  - score: A confidence score between 0.0 and 1.0 for the classification
  """
  @primary_key false
  embedded_schema do
    field(:class, Ecto.Enum, values: [:spam, :not_spam])
    field(:reason, :string)
    field(:score, :float)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_number(:score,
      greater_than_or_equal_to: 0.0,
      less_than_or_equal_to: 1.0
    )
  end
end

is_spam? = fn text ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: SpamPredicition,
    max_retries: 3,
    messages: [
      %{
        role: "user",
        content: """
        You purpose is to classify customer support emails as either spam or not.
        This is for a clothing retailer business.
        They sell all types of clothing.

        Classify the following email: #{text}
        """
      }
    ]
  )
end

is_spam?.("Hello I am a Nigerian prince and I would like to send you money")

# => {:ok, %SpamPrediction{class: :spam, reason: "Nigerian prince email scam", score: 0.98}}
Simply create an ecto schema, optionally provide a @doc to the schema definition which we pass down to the LLM, then make a call to Instructor.chat_completion/1 with context about the task you'd like the LLM to complete.
You can also provide a validate_changeset/1 function via the use Instructor.Validator which will provide a set of code level ecto changeset validations. You can use this in conjunction with max_retries: 3 to automatically, iteratively go back and forth with the LLM up to n times with any validation errors so that it has a chance to fix them.
Curious to learn more? Unsure of how you'd use this? Check out our extensive set of tutorials
	Tutorial - Basic Usage & Features
	Text Classification
	Extracting Action Items from Meeting Transcriptions
	Extracting Explorer.DataFrames from text


  
    
  
  Configuration


To configure the default OpenAI adapter you can set the configuration,
config :openai, api_key: "sk-........"
config :openai, http_options: [recv_timeout: 10 * 60 * 1000]
To use a local LLM, you can install and run llama.cpp server and tell instructor to use it,
config :instructor, adapter: Instructor.Adapters.Llamacpp

  
    
  
  Installation


In your mix.exs,
def deps do
  [
    {:instructor, "~> 0.0.3"}
  ]
end

  
    
  
  TODO


	[ ] llamacpp adapter broken, needs to support openai input/output API	[ ] GBNF should enforce required properties on objects, currently they're optional.
	[ ] GBNF limit the number of digits in number tokens -- small models can sometimes run off to infinit digits


	[ ] Add llm_validator
	[ ] Logging for Distillation / Finetuning
	[ ] Add a Bumblebee adapter
	[ ] Add llamacpp_ex adapter
	[x] Support Streaming	[ ] Verify schemaless support {:array, %{name: :string}}
	[ ] Support typespec style support for array streaming [MySchema]


	[ ] Optional/Maybe types
	[ ] When changeset fails to cast, it should show the underlying data (better error messages)
	[ ] Add Livebook Tutorials, include in Hexdocs	[x] Text Classification
	[ ] Self Critique
	[ ] Image Extracting Tables
	[ ] Moderation
	[x] Citations
	[ ] Knowledge Graph
	[ ] Entity Resolution
	[ ] Search Queries
	[ ] Query Decomposition
	[ ] Recursive Schemas
	[x] Table Extraction
	[x] Action Item and Dependency Mapping
	[ ] Multi-File Code Generation
	[ ] PII Data Sanitizatiommersed


	[x] Update hexdocs homepage to include example for tutorial
	[ ] Setup Github CI for testing, add badge to README


  
    
  
  Blog Posts


	[ ] Why structured prompting?
  Meditations on new HCI.
  Finally we have software that can understand text. f(text) -> text.
  This is great, as it gives us a new domain, but the range is still text.
  While we can use string interpolation to map Software 1.0 into f(text), the outputs are not interoperable with Software 1.0.
  Hence why UXs available to us are things like Chatbots as our users have to interpret the output.
  Instructor, structure prompting, gives use f(text) -> ecto_schema.
  Schemas are the lingua franca of Software 1.0.
  With Instrutor we can now seamlessly move back and forth between Software 1.0 and Software 2.0.
  Now we can maximally leverage AI...

	[ ] From GPT-4 to zero-cost production - Distilation, local-llms, and the cost structure of AI.
  ... 😘





  

    
Tutorial
    

Mix.install(
  [
    # {:instructor, ">= 0.0.2"}
    {:instructor, path: "/Users/thomas/code/instructor_ex"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI
    ],
    openai: [
      api_key: System.fetch_env!("LB_OPENAI_API_KEY"),
      http_options: [recv_timeout: 10 * 60 * 1000]
    ]
  ]
)

  
    
  
  Introduction


Instructor is a library to do structured prompting with OpenAI and open source LLMs. While the idea is pretty simple, through this and the other examples you'll realize how powerful a concept this is.
So first off, what is structure prompting?
What if the LLM returned data conforming to a complicated nested schema that your code knows how to work with? Well, that's structure prompting. It's a way of cohercing the LLM to producing it's response in a known format that your downstream code can handle. In the case of Instructor, we use Ecto to provide those schemas. Good old Ecto, something you're already familiar with.
So, without further ado, let's take define a schema and take it for a spin!
defmodule Politician do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  A description of United States Politicians and the offices that they held,

  ## Fields:
  - first_name: Their first name
  - last_name: Their last name
  - offices_held:
    - office: The branch and position in government they served in
    - from_date: When they entered office or null
    - until_date: The date they left office or null
  """
  @primary_key false
  embedded_schema do
    field(:first_name, :string)
    field(:last_name, :string)

    embeds_many :offices_held, Office, primary_key: false do
      field(:office, Ecto.Enum,
        values: [:president, :vice_president, :govenor, :congress, :senate]
      )

      field(:from_date, :date)
      field(:to_date, :date)
    end
  end
end
{:module, Politician, <<70, 79, 82, 49, 0, 0, 17, ...>>,
 [__schema__: 1, __schema__: 1, __schema__: 1, __schema__: 1, __schema__: 2, __schema__: 2, ...]}
Great, we have our schema describing politicans and the offices they held. Let's notice a few things that may stand out from regular Ecto usage. First, since there is no database backing the schema, it doesn't make sense to give it a primary_key. This also makes sense because there is no sensible value for the LLM to respond with.
Also we use a @doc on the schema. This isn't just for documentation purposes of the tutorial. Instructor will take any @doc tag and provide it to the LLM. Generally you'll want to use this to provide semantic descriptions of the fields and general context to the LLM to ensure you get the outputs you want. In our case we want to push the LLM to understand that we are only considering American politicians.
So, let's try asking the LLM to give us some politicians.
Instructor.chat_completion(
  model: "gpt-3.5-turbo",
  response_model: Politician,
  messages: [
    %{
      role: "user",
      content:
        "Who won the American 2020 election and what offices have they held over their career?"
    }
  ]
)
{:ok,
 %Politician{
   first_name: "Joe",
   last_name: "Biden",
   offices_held: [%Politician.Office{office: :president, from_date: ~D[2021-01-20], to_date: nil}]
 }}
Amazing, right? Using nothing more than one of the top libraries in Elixir, Ecto, we were able to get structured output from our LLM. The data returned is ready to be processed by our regular Elixir code. Instructor supports all field types that you can express in Ecto, including embedded and associated schemas.
It's almost as if the LLM inputted the data into a Phoenix Form. All the utilities that you use to process that kind of data, you can use to process the outputs of Instructor.
One of the superpowers of this is that since we're just using changesets under the hood, you can use the same validations that you would use elsewhere in your app. Let's look at that in the next section.

  
    
  
  Validations


Instructor provides a lightweight behavior where you can define a callback function that we will call to validate the data returned by the LLM using Ecto changesets. There is nothing fancy to this API. It's just a changeset in and a changeset out.
defmodule NumberSeries do
  use Ecto.Schema
  use Instructor.Validator

  @primary_key false
  embedded_schema do
    field(:series, {:array, :integer})
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_length(:series, min: 10)
    |> Ecto.Changeset.validate_change(:series, fn
      field, values ->
        if Enum.sum(values) |> rem(2) == 0 do
          []
        else
          [{field, "The sum of the series must be even"}]
        end
    end)
  end
end
{:module, NumberSeries, <<70, 79, 82, 49, 0, 0, 18, ...>>, {:validate_changeset, 1}}
In this albeit contrived example, we're going to get the LLM to return a series of numbers and validate whether it has at least 10 numbers and that the sum of the series is even.
When we ask for fewer than ten numbers, Instructor will return an error tuple with a change set that is invalid.
{:error, changeset} =
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: NumberSeries,
    messages: [
      %{role: "user", content: "Give me the first 5 integers"}
    ]
  )

# Render our the errors down to strings.
errors =
  Ecto.Changeset.traverse_errors(changeset, fn {msg, opts} ->
    Regex.replace(~r"%{(\w+)}", msg, fn _, key ->
      opts |> Keyword.get(String.to_existing_atom(key), key) |> to_string()
    end)
  end)

{changeset.changes, errors}
{%{series: [1, 2, 3, 4, 5]},
 %{series: ["The sum of the series must be even", "should have at least 10 item(s)"]}}
Now the beauty of this is that since we have human readable errors from our validations, we can just turn around and pass those back into the LLM to get it to fix its own errors.
Instructor provides a convenience parameter, max_retries for you in the initial call which will retry against the validations up to n times.
Instructor.chat_completion(
  model: "gpt-3.5-turbo",
  response_model: NumberSeries,
  max_retries: 10,
  messages: [
    %{role: "user", content: "Give some random integers"}
  ]
)

23:59:13.641 [debug] Retrying LLM call for Elixir.NumberSeries...

{:ok, %NumberSeries{series: [2, 4, 6, 8, 10, 12, 14, 16, 18, 20]}}



  

    
Text Classification
    

Mix.install(
  [
    {:instructor, "~> 0.0.2"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI
    ],
    openai: [
      api_key: System.fetch_env!("LB_OPENAI_API_KEY"),
      http_options: [recv_timeout: 10 * 60 * 1000]
    ]
  ]
)

  
    
  
  Motivation


Text classification is a common task in NLP and broadly applicable across software. Whether it be spam detection, or support ticket categorization, NLP is at the core. Historically, this required training custom, bespoke models that required collecting thousands of pre-labeled examples. With LLMs a lot of this knowledge is already encoded into the model. With proper instruction and guiding the output to a known set of classifications using GPT you can be up and running with a text classification model in no time.
Hell, you can even use instructor to help generate the training set to train your own more efficient model. But let's not get ahead of ourselves, there's more on that later in the tutorials.

  
    
  
  Binary Text Classification


Spam detection is a classic example of binary text classification. It's as simple as returning a true / false of whether an example is in the class. This is pretty trivial to implement in instructor.
defmodule SpamClassification do
  use Ecto.Schema

  @doc """
  A classification of whether or not a provided email is spam or not
  """
  @primary_key false
  embedded_schema do
    field(:is_spam?, :boolean)
  end
end

is_spam? = fn text ->
  {:ok, %{is_spam?: result}} =
    Instructor.chat_completion(
      model: "gpt-3.5-turbo",
      response_model: SpamClassification,
      messages: [
        %{
          role: "user",
          content: "Classify the following text as spam/not_spam: #{text}"
        }
      ]
    )

  result
end

is_spam?.("Hello I am a Nigerian prince and I would like to send you money")
true

  
    
  
  Multi-Class Text Classification


We don't have to stop just at a boolean inclusion, we can also easily extend this idea to multiple categories or classes that we can classify the text into. In this example, let's consider classifying support emails. We want to know whether it's a general_inquiry, billing_issue, or a technical_issue perhaps it rightly fits in multiple classes. This can be useful if we want to cc' specialized support agents when intersecting customer issues occur
We can leverage Ecto.Enum to define a schema that restricts the LLM output to be a list one of those values. We can also provide a @doc description to help guide the LLM with the semantic understanding of what these classifications ought to represent.
defmodule EmailClassifications do
  use Ecto.Schema

  @doc """
  A classification of a customer support email.

  technical_issue - whether the user is having trouble accessing their account.
  billing_issue - whether the customer is having trouble managing their billing or credit card
  general_inquiry - all other issues
  """
  @primary_key false
  embedded_schema do
    field(:tags, {:array, Ecto.Enum},
      values: [:general_inquiry, :billing_issue, :technical_issue]
    )
  end
end

classify_email = fn text ->
  {:ok, %{tags: result}} =
    Instructor.chat_completion(
      model: "gpt-3.5-turbo",
      response_model: EmailClassifications,
      messages: [
        %{
          role: "user",
          content: "Classify the following text: #{text}"
        }
      ]
    )

  result
end

classify_email.("My account is locked and I can't access my billing info.")
[:technical_issue, :billing_issue]



  

    
Extracting Action Items from Meeting Transcripts
    

Mix.install(
  [
    # {:instructor, "~> 0.0.2"}
    {:instructor, path: "/Users/thomas/code/instructor_ex"},
    {:kino, "~> 0.12.0"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI
    ],
    openai: [
      api_key: System.fetch_env!("LB_OPENAI_API_KEY"),
      http_options: [recv_timeout: 10 * 60 * 1000]
    ]
  ]
)

  
    
  
  Motivation


This example shouldn't be foreign to any of you. I'm sure each day you log in, join a Zoom call, and have a meeting about what y'all are going to do this upcoming week. It is then the work of some product manager to translate this into a JIRA board so that you can track your progress throughout the week.
The bane of most engineers existence... We can automate this.

  
    
  
  The Schema


Let's start by defining a schema for the tickets and the subtasks that might exist within them. There will be a priority, description and a set of dependencies between the tasks for the tickets.
defmodule MeetingNotes do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  Tickets correctly resolved from a meeting transcription.

  Tickets have a name, priority, useful description, and assignees.
  They may also have subtasks that share the same id spaces as the tickets.
  A ticket also may have references to Tickets and Subtasks that are a blocking dependency
  for completing the ticket.
  """
  @primary_key false
  embedded_schema do
    embeds_many :tickets, Ticket do
      field(:name, :string)
      field(:description, :string)
      field(:priority, Ecto.Enum, values: [:high, :medium, :low])
      field(:assignees, {:array, :string})

      embeds_many :subtasks, SubTasks do
        field(:name, :string)
      end

      field(:dependencies, {:array, :binary_id})
    end
  end
end
{:module, MeetingNotes, <<70, 79, 82, 49, 0, 0, 17, ...>>,
 [__schema__: 1, __schema__: 1, __schema__: 1, __schema__: 1, __schema__: 2, __schema__: 2, ...]}
One thing interesting to note here is that although we could use proptorecto associations, in this case it's rather tedious. Since our dependencies can either be on tickets or on subtasks, you would typically have to have a polymorphic relationship.
While Instructor fully supports doing that, we can instead just create embedded schemas for the subtasks and define that the subtasks and the tickets share the same ID space in our doc comment. This will steer the LLM to produce foreign key relations in a polymorphic way.
One also might choose to solve this modeling problem by denormalizing the tickets and subtasks into its own task type and get the LLM to not only have the dependencies listed but also any subtasks that a ticket might have through association instead of embedding.
Either way works, but in this example we prefer the embedded method because it produces a prettier output and we can lean on the LLM a little more.
Now, let's extract some JIRA tickets.
generate_tickets = fn transcript ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: MeetingNotes,
    messages: [
      %{role: "system", content: "The following is a transcript of a meeting..."},
      %{
        role: "user",
        content: "Create the action items for the following transcript: #{transcript}"
      }
    ]
  )
end

{:ok, %{tickets: tickets}} =
  generate_tickets.("""
    Alice: Hey team, we have several critical tasks we need to tackle for the upcoming release. First, we need to work on improving the authentication system. It's a top priority.
    
    Bob: Got it, Alice. I can take the lead on the authentication improvements. Are there any specific areas you want me to focus on?
    
    Alice: Good question, Bob. We need both a front-end revamp and back-end optimization. So basically, two sub-tasks.
    
    Carol: I can help with the front-end part of the authentication system.
    
    Bob: Great, Carol. I'll handle the back-end optimization then.
    
    Alice: Perfect. Now, after the authentication system is improved, we have to integrate it with our new billing system. That's a medium priority task.
    
    Carol: Is the new billing system already in place?
    
    Alice: No, it's actually another task. So it's a dependency for the integration task. Bob, can you also handle the billing system?
    
    Bob: Sure, but I'll need to complete the back-end optimization of the authentication system first, so it's dependent on that.
    
    Alice: Understood. Lastly, we also need to update our user documentation to reflect all these changes. It's a low-priority task but still important.
    
    Carol: I can take that on once the front-end changes for the authentication system are done. So, it would be dependent on that.
    
    Alice: Sounds like a plan. Let's get these tasks modeled out and get started.
  """)
{:ok,
 %MeetingNotes{
   tickets: [
     %MeetingNotes.Ticket{
       id: "T1",
       name: "Improving authentication system",
       description: "Work on improving the authentication system",
       priority: :high,
       assignees: ["Bob"],
       subtasks: [
         %MeetingNotes.Ticket.SubTasks{id: "T1.1", name: "Front-end revamp"},
         %MeetingNotes.Ticket.SubTasks{id: "T1.2", name: "Back-end optimization"}
       ],
       dependencies: []
     },
     %MeetingNotes.Ticket{
       id: "T2",
       name: "Integrating authentication system with billing system",
       description: "Integrate the improved authentication system with the new billing system",
       priority: :medium,
       assignees: ["Bob"],
       subtasks: [],
       dependencies: ["T1.2"]
     },
     %MeetingNotes.Ticket{
       id: "T3",
       name: "Update user documentation",
       description: "Update user documentation to reflect the changes",
       priority: :low,
       assignees: ["Carol"],
       subtasks: [],
       dependencies: ["T1.1"]
     },
     %MeetingNotes.Ticket{
       id: "T4",
       name: "Handle front-end part of authentication system",
       description: "Help with the front-end part of the authentication system",
       priority: :medium,
       assignees: ["Carol"],
       subtasks: [],
       dependencies: []
     },
     %MeetingNotes.Ticket{
       id: "T5",
       name: "Handle back-end optimization of authentication system",
       description: "Handle the back-end optimization of the authentication system",
       priority: :high,
       assignees: ["Bob"],
       subtasks: [],
       dependencies: []
     },
     %MeetingNotes.Ticket{
       id: "T6",
       name: "Handle billing system",
       description: "Handle the new billing system",
       priority: :medium,
       assignees: ["Bob"],
       subtasks: [],
       dependencies: ["T5"]
     }
   ]
 }}
The results look good. We can use Kino to then render out all the dependencies of our tickets.
generate_tickets_diagram = fn tickets ->
  subtasks = Enum.flat_map(tickets, & &1.subtasks)
  all_tasks = subtasks ++ tickets

  ticket_nodes =
    tickets
    |> Enum.map_join("\n", fn t ->
      subtask_relations =
        t.subtasks
        |> Enum.map_join("\n", fn st ->
          """
          "#{t.name}" ||--o| "#{st.name}" : "Has Subtask"
          """
        end)

      dependency_relations =
        t.dependencies
        |> Enum.map_join("\n", fn d ->
          dt = all_tasks |> Enum.find(&(&1.id == d))

          if dt do
            """
            "#{t.name}" ||--o| "#{dt.name}" : "Depends on"
            """
          else
            ""
          end
        end)

      """
        "#{t.name}" {
          priority #{t.priority}
          assignees #{Enum.join(t.assignees, ", ")}
        }

        #{subtask_relations}
        #{dependency_relations}
      """
    end)

  subtask_nodes =
    tickets
    |> Enum.flat_map(& &1.subtasks)
    |> Enum.map_join("\n", fn st ->
      """
        "#{st.name}"
      """
    end)

  Kino.Mermaid.new("""
    erDiagram
      #{ticket_nodes}
      #{subtask_nodes}
  """)
end

generate_tickets_diagram.(tickets)
  erDiagram
      "Improving authentication system" {
    priority high
    assignees Bob
  }

  "Improving authentication system" ||--o| "Front-end revamp" : "Has Subtask"

"Improving authentication system" ||--o| "Back-end optimization" : "Has Subtask"

  

  "Integrating authentication system with billing system" {
    priority medium
    assignees Bob
  }

  
  "Integrating authentication system with billing system" ||--o| "Back-end optimization" : "Depends on"


  "Update user documentation" {
    priority low
    assignees Carol
  }

  
  "Update user documentation" ||--o| "Front-end revamp" : "Depends on"


  "Handle front-end part of authentication system" {
    priority medium
    assignees Carol
  }

  
  

  "Handle back-end optimization of authentication system" {
    priority high
    assignees Bob
  }

  
  

  "Handle billing system" {
    priority medium
    assignees Bob
  }

  
  "Handle billing system" ||--o| "Handle back-end optimization of authentication system" : "Depends on"


      "Front-end revamp"

  "Back-end optimization"





  

    
Converting Text into Explorer DataFrames
    

Mix.install(
  [
    # {:instructor, "~> 0.0.2"}
    {:instructor, path: "/Users/thomas/code/instructor_ex"},
    {:explorer, "~> 0.7.2"},
    {:kino, "~> 0.12.0"},
    {:kino_explorer, "~> 0.1.13"}
  ],
  config: [
    instructor: [
      adapter: Instructor.Adapters.OpenAI
    ],
    openai: [
      api_key: System.fetch_env!("LB_OPENAI_API_KEY"),
      http_options: [recv_timeout: 10 * 60 * 1000]
    ]
  ]
)

  
    
  
  Motivation


Sometimes we come across text that can be naturally extracted into structured data. Although we don't necessarily know how many tables, and what the schemas are represented in the data. Using Instructor we can let the LLM determine the tables and schemas for us, pulling the results into a dataframe which we can then use for further analysis.

  
    
  
  The Schema


Notice from this example we want resulting data structure to have data frames. That's not a default Ecto type so we're gonna have to do things a little different here.
Luckily, Ecto allows us to define our own custom types in which we can override how the values are casted. And Instructor provides us a behavior which tells us how to represent it in a JSONSchema so the LLM can understand it.
In our case, we're going to create a type for a table of data, which will be represented as a CSV string with respect to the LLM. Our custom type will then parse the string when it is casted by Ecto and parse the CSV into an Explorer.DataFrame.
defmodule Ecto.CSVDataFrame do
  use Ecto.Type
  use Instructor.EctoType

  def type, do: :string

  def cast(csv_str) when is_binary(csv_str) do
    df = Explorer.DataFrame.load_csv!(csv_str)
    {:ok, df}
  end

  def cast(%Explorer.DataFrame{} = df), do: {:ok, df}
  def cast(_), do: :error

  def to_json_schema(),
    do: %{
      type: "string",
      description: "A CSV representation of a data table"
    }

  def dump(x), do: {:ok, x}
  def load(x), do: {:ok, x}
end
{:module, Ecto.CSVDataFrame, <<70, 79, 82, 49, 0, 0, 12, ...>>, {:load, 1}}
Now that we have our data type, we can use it in our embedded ecto schema, just as we would have in any other example. If everything works correctly, we should get some data frames out.
defmodule Database do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  The extracted database will contain one or more tables with data as csv formatted with ',' delimiters
  """
  @primary_key false
  embedded_schema do
    embeds_many :tables, DataFrame, primary_key: false do
      field(:name, :string)
      field(:data, Ecto.CSVDataFrame)
    end
  end
end
{:module, Database, <<70, 79, 82, 49, 0, 0, 16, ...>>,
 [__schema__: 1, __schema__: 1, __schema__: 1, __schema__: 1, __schema__: 2, __schema__: 2, ...]}
Perfect. And now let's test it out on some example text.
extract_df = fn text ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: Database,
    messages: [
      %{
        role: "system",
        content: """
        Map this data into one or more dataframes and correctly define the columns and data
        """
      },
      %{
        role: "user",
        content: "#{text}"
      }
    ]
  )
end

{:ok, db} =
  extract_df.("""
    My name is John and I am 25 years old. I live in
    New York and I like to play basketball. His name is
    Mike and he is 30 years old. He lives in San Francisco
    and he likes to play baseball. Sarah is 20 years old
    and she lives in Los Angeles. She likes to play tennis.
    Her name is Mary and she is 35 years old.
    She lives in Chicago.
    
    On one team 'Tigers' the captain is John and there are 12 players.
    On the other team 'Lions' the captain is Mike and there are 10 players.
  """)

Kino.Layout.tabs(
  for table <- db.tables do
    {table.name, table.data}
  end
)



  

    
Instructor 
    



      
Instructor.ex is a spiritual port of the great Instructor Python Library by @jxnlco, check out his talk on youtube.
This library brings structured prompting to LLMs. Instead of receiving text as output, Instructor will coax the LLM to output valid JSON that maps directly to the provided Ecto schema.
If the LLM fails to do so, or provides values that do not pass your validations, it will provide you utilities to automatically retry with the LLM to correct errors.
By default it's designed to be used with the OpenAI API, however it provides an extendable adapter behavior to work with ggerganov/llama.cpp and Bumblebee (Coming Soon!).
At its simplest, usage is pretty straightforward,
defmodule SpamPrediction do
  use Ecto.Schema
  use Instructor.Validator

  @doc """
  ## Field Descriptions:
  - class: Whether or not the email is spam
  - reason: A short, less than 10 word rationalization for the classification
  - score: A confidence score between 0.0 and 1.0 for the classification
  """
  @primary_key false
  embedded_schema do
    field(:class, Ecto.Enum, values: [:spam, :not_spam])
    field(:reason, :string)
    field(:score, :float)
  end

  @impl true
  def validate_changeset(changeset) do
    changeset
    |> Ecto.Changeset.validate_number(:score,
      greater_than_or_equal_to: 0.0,
      less_than_or_equal_to: 1.0
    )
  end
end

is_spam? = fn text ->
  Instructor.chat_completion(
    model: "gpt-3.5-turbo",
    response_model: SpamPredicition,
    max_retries: 3,
    messages: [
      %{
        role: "user",
        content: """
        You purpose is to classify customer support emails as either spam or not.
        This is for a clothing retailer business.
        They sell all types of clothing.

        Classify the following email: #{text}
        """
      }
    ]
  )
end

is_spam?.("Hello I am a Nigerian prince and I would like to send you money")

# => {:ok, %SpamPrediction{class: :spam, reason: "Nigerian prince email scam", score: 0.98}}
Simply create an ecto schema, optionally provide a @doc to the schema definition which we pass down to the LLM, then make a call to Instructor.chat_completion/1 with context about the task you'd like the LLM to complete.
You can also provide a validate_changeset/1 function via the use Instructor.Validator which will provide a set of code level ecto changeset validations. You can use this in conjunction with max_retries: 3 to automatically, iteratively go back and forth with the LLM up to n times with any validation errors so that it has a chance to fix them.
Curious to learn more? Unsure of how you'd use this? Check out our extensive set of tutorials
	Tutorial - Basic Usage & Features
	Text Classification
	Extracting Action Items from Meeting Transcriptions
	Extracting Explorer.DataFrames from text


  
    
  
  Configuration


To configure the default OpenAI adapter you can set the configuration,
config :openai, api_key: "sk-........"
config :openai, http_options: [recv_timeout: 10 * 60 * 1000]
To use a local LLM, you can install and run llama.cpp server and tell instructor to use it,
config :instructor, adapter: Instructor.Adapters.Llamacpp

        chat_completion(params)

      


        Create a new chat completion for the provided messages and parameters.

      Link to this function
    
    cast_all(schema, params)

      Link to this function
    
    chat_completion(params)

          @spec chat_completion(Keyword.t()) ::
  {:ok, Ecto.Schema.t()} | {:error, Ecto.Changeset.t()} | {:error, String.t()}


      


Create a new chat completion for the provided messages and parameters.
The parameters are passed directly to the LLM adapter.
By default they shadow the OpenAI API parameters.
For more information on the parameters, see the OpenAI API docs.
Additionally, the following parameters are supported:
	:response_model - The Ecto schema to validate the response against.
	:max_retries - The maximum number of times to retry the LLM call if it fails, or does not pass validations.             (defaults to `0`)



  
    
  
  Examples


  iex> Instructor.chat_completion(%{
  ...>   model: "gpt-3.5-turbo",
  ...>   response_model: Instructor.Demos.SpamPrediction,
  ...>   messages: [
  ...>     %{
  ...>       role: "user",
  ...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
  ...>     }
  ...> })
  {:ok,
  %Instructor.Demos.SpamPrediction{
      class: :spam
      score: 0.999
  }}

Behavior for Instructor.Adapter.

    chat_completion(map)

          @callback chat_completion(map()) :: any()


      



  


        

      



  

    
Instructor.Adapters.Llamacpp 
    



      
Runs against the llama.cpp server. To be clear this calls the llamacpp specific
endpoints, not the open-ai compliant ones.
You can read more about it here:
  https://github.com/ggerganov/llama.cpp/tree/master/examples/server

        chat_completion(params, config \\ nil)

      


        Run a completion against the llama.cpp server, not the open-ai compliant one.
This gives you more specific control over the grammar, and the ability to
provide other parameters to the specific LLM invocation.

      Link to this function
    
    chat_completion(params, config \\ nil)

Run a completion against the llama.cpp server, not the open-ai compliant one.
This gives you more specific control over the grammar, and the ability to
provide other parameters to the specific LLM invocation.
You can read more about the parameters here:
  https://github.com/ggerganov/llama.cpp/tree/master/examples/server

  
    
  
  Examples


  iex> Instructor.chat_completion(%{
  ...>   model: "mistral-7b-instruct",
  ...>   messages: [
  ...>     %{ role: "user", content: "Classify the following text: Hello I am a Nigerian prince and I would like to send you money!" },
  ...>   ],
  ...>   response_model: response_model,
  ...>   temperature: 0.5,
  ...> })

Documentation for Instructor.Adapters.OpenAI.

          @callback to_json_schema() :: map()

          Generates a JSON Schema from an Ecto schema.

  Generates a JSON Schema from an Ecto schema.
  Note: This will output a correct JSON Schema for the given Ecto schema, but
  it will not necessarily be optimal, nor support all Ecto types.

By default you'll get whatever OpenAI returns.
This behavior provides a hook for you to critique the response using standard ecto changesets validations.
This can be used in conjuction with the :max_retries parameter to Instructor.chat_completion/1 to retry the completion until it passes your validation.

  
    
  
  Examples


  defmodule Instructor.Demos.SpamPrediction do
use Ecto.Schema

@primary_key false
schema "spam_prediction" do
    field :class, :string
    field :score, :float
end

@impl true
def validate_changeset(changeset) do
    changeset
    |> validate_number(:score, less_than_or_equal_to: 1.0)
end
  end
  iex> Instructor.chat_completion(%{
  ...>   model: "gpt-3.5-turbo",
  ...>   response_model: Instructor.Demos.SpamPrediction,
  ...>   max_retries: 1,
  ...>   messages: [
  ...>     %{
  ...>       role: "user",
  ...>       content: "Classify the following text: Hello, I am a Nigerian prince and I would like to give you $1,000,000."
  ...>     }
  ...> })
  {:error, %Ecto.Changeset{
  action: nil,
  changes: %{},
  errors: [
      score: {"is invalid", [validation: :number, validation_opts: [less_than_or_equal_to: 1.0]]}
  ],
  data: %Instructor.Demos.SpamPrediction{
      class: nil,
      score: nil
  },
  valid?: false
  }}

          @callback validate_changeset(Ecto.Changeset.t(), Map.t()) :: Ecto.Changeset.t()

